PROGRAMA MASTERMI.LOG

;COMENTARIO: Los siguientes procedimientos sirven para definir las distintas melodías y sonidos que usa el programa

Para fallo

tono 100 10

fin

Para tetoca

tono 1024 10

Fin

Para perdiste

tono 341 50 tono 320 50 tono 304 50 tono 288 50

Fin

Para ganaste

Repite 3 [tono 512 10 tono 640 10 tono 768 10 tono 1024 20]

Fin

Para obertura

Repite 3 [tono 512 10 tono 640 10 tono 768 10 tono 1024 20 tono 768 10 tono 1024 20]

Fin

Para salida

tono 1024 20   tono 768 10 tono 768 10 tono 810 20 tono 768 20 espera 20 tono 960 20   tono 1024 10

Fin

;COMENTARIO: Los dos siguientes procedimientos sirven para solicitar al usuario los 4 números que van a formar su jugada, detectan si hay algún error en la introducción de dichos números y dibujan cuadrados con los colores asociados a los mismos. La jugada se guarda en la variable :jugada y la variable cf es un contador que sirve para situar los cuadraditos en sus columnas correspondientes.

Para datos

PonFondoTexto 5 PonColorTexto  12

Escribe [Introduce 4 números comprendidos entre 1 y 6:]

PonFondoTexto 16 PonColorTexto 1

Haz "jugada [] Haz "cf 1

Repite 4 [petición]

Escribe []

Fin

Para petición

Haz "dato LC

Si No Número? :dato [fallo petición Alto]

Si O :dato<1 :dato>6 [fallo petición Alto]

(Teclea :dato car 32)

Haz "jugada PonÚltimo :dato :jugada

SL PonPos Lista -35+20*:cf 70-15*:contadordejugadas BL PonRumbo 0 botón :dato

Haz "cf :cf+1

Fin

;COMENTARIO: El procedimiento siguiente genera la combinación de colores que hay que adivinar.

Para solución

Haz "solución []

Repite 4 [Haz "solución Ponúltimo 1+Azar 6 :solución]

Fin

;COMENTARIO: El procedimiento siguiente indica el número de aciertos de color y posición simultáneamente. Estos aciertos se guardan en la variable :contador1.

Para comparar1

Haz "contador1 0

Si Elemento 1 :jugada = Elemento 1 :solución [Haz "contador1 :contador1 + 1]

Si Elemento 2 :jugada = Elemento 2 :solución [Haz "contador1 :contador1 + 1]

Si Elemento 3 :jugada = Elemento 3 :solución [Haz "contador1 :contador1 + 1]

Si Elemento 4 :jugada = Elemento 4 :solución [Haz "contador1 :contador1 + 1]

Fin

;COMENTARIO: En este procedimiento se duplica la variable :jugada para hacer modificaciones en ella sin perder el valor inicial (esto no es realmente necesario pero en un principio pareció adecuado), además se inicializa la variable que guardará el número de aciertos de color pero no de posición.

Para variables

Haz "j :jugada

Haz "contador2 0

Fin

;COMENTARIO: En este procedimiento se efectúa el segundo tipo de comparación. La idea es averiguar cuántas veces aparece el primer elemento de :jugada en la propia jugada y en la solución. El número de aciertos será la menor de esas dos cantidades y se almacena en contador2. Después hay que eliminar el elemento comparado de la lista que contiene la jugada y repetir el proceso hasta que ya no queden elementos.

Para comparar2

Haz "Primer Primero :j

Haz "v1 Veces :Primer :j

Haz "v2 Veces :Primer :solución

Si :v1<:v2 [Haz "contador2 :contador2 + :v1] [Haz "contador2 :contador2 + :v2]

Repite :v1 [Haz "j Elimina :Primer :j]

Si :j = [] [Alto]

comparar2

Fin

;COMENTARIO: El siguiente procedimiento define de forma unitaria la jugada del usuario: en primer lugar averigua si ya lleva más de 10 intentos, en cuyo caso la partida termina; a continuación pide la jugada, efectúa las dos comparaciones y nos da los mensajes con los aciertos. Se han preparado varios mensajes para que la expresión gramatical sea correcta. Se dibuja la respuesta en forma gráfica, se averigua si se ha acertado y en caso contrario se reinicia el proceso.

Para mijugada

Haz "contadordejugadas :contadordejugadas + 1

Si :contadordejugadas > 10 [perdiste PonCT 10 (Escribe [¡Lo siento! Has perdido. La solución era: ] car 32 :solución) PonCT 1 dibujasolución Alto]

datos 

comparar1

variables

comparar2

dibujarespuesta

PonCT 6

Si :contador2 = 1 [Mensaje1] [Mensaje2]

PonCT 1

Si :contador1 = 4 [ganaste PonCT 13 Escribe [¡Enhorabuena! Has acertado la combinación.] PonCT 1 dibujasolución Alto]

tetoca

mijugada

Fin

;COMENTARIO: Éste es el procedimiento principal: se introduce la melodía de entrada, se organiza la presentación de la pantalla, se genera la solución, se inicializa el contador de jugadas, se pide al usuario que inicie la partida y una vez finalizada se le pregunta si quiere volver a empezar.

Para MasterMind

obertura

presentación

info

solución

Haz "contadordejugadas 0

mijugada

otravez

Fin

;COMENTARIO: los dos procedimientos siguientes son los mensajes que indican los aciertos de cada jugada.

Para Mensaje2

Si :contador2=0 [Escribe [No has acertado ninguno] Alto]

Si No :contador1=1 [(Escribe [Has acertado] :contador2 [colores, de los que] :contador1 [están en posición correcta.]) Alto]

(Escribe [Has acertado] :contador2 [colores, de los que] :contador1 [está en posición correcta.])

Fin

Para Mensaje1

Si :contador1=1 [(Escribe [Has acertado] :contador2 [color, y está en posición correcta.]) Alto]

(Escribe [Has acertado] :contador2 [color, y no está en posición correcta.]

Fin

;COMENTARIO: El siguiente procedimiento prepara la pantalla para una buena presentación del programa.

Para presentación

BorraPantalla BorraTexto

CierraVentana "TRABAJO

PonPosicionVentana "GRAFICOS [0 0]

PonTamañoVentana "GRAFICOS [640 300]

PonPosicionVentana "TEXTOS [0 300]

PonTamañoVentana "TEXTOS [640 113]

OT SL PonPos [-240 60] PonCl 13 BL

Repite 2 [Av 40 GD 90 Av 200 GD 90]

SL Av 20 GD 90 Av 20 BL

Rellena

SL Re 20 GI 90 Re 20 BL

PonCL 1 PonGrosor 3

Repite 2 [Av 40 GD 90 Av 200 GD 90]

PonGrosor 1 PonCL 12 PonTamañoTipo [0.8 0.8]

PonTipo "GOTICA

SL PonPos [-230 70] PonRumbo 90 BL

Rotula "MASTERMIND

PonCL 1

SL PonPos [-10 80] BL

PonTipo "ROMAN PonTamañoTipo [0.4 0.4] PonCL 6

Rotula "JUGADAS

SL PonPos [65 65] BL PonY -90

SL PonPos [75 80] BL  PonCL 4

Rotula "ACIERTOS

SL PonPos [160 80] BL PonCL 222

Rotula "SOLUCION

SL PonPos [155 65] BL PonY -90

Haz "numjug 1

PonCL 6 PonTamañoTipo [0.3 0.3]

Repite 10 [SL PonPos Lista -30 70-15*:numjug BL Rotula :numjug Haz "numjug :numjug+1]

Fin

;COMENTARIO: Los siguientes procedimientos generan los cuadraditos de colores y el blanco y el negro.

Para botón :x

Si :x=1 [PonCL 10]

Si :x=2 [PonCL 11]

Si :x=3 [PonCL 12]

Si :x=4 [PonCL 13]

Si :x=5 [PonCL 234]

Si :x=6 [PonCL 3]

Repite 4 [Av 10 GD 90]

SL Av 5 GD 90 Av 5 BL

Rellena

SL Re 5 GI 90 Re 5 BL

PonCL 1

Fin

Para botónb

Repite 4 [Av 10 GD 90]

Fin

Para botónn

Repite 4 [Av 10 GD 90]

SL Av 5 GD 90 Av 5 BL

Rellena

SL Re 5 GI 90 Re 5 BL

Fin

;COMENTARIO: El siguiente procedimiento incorpora la información necesaria para jugar. Ésta se presenta en la pantalla gráfica.

Para info

SL PonPos [-220 -90] PonRumbo 0 BL

PonGrosor 3

Repite 2 [Av 140 GD 90 Av 160 GD 90]

PonGrosor 1

SL PonPos [-200 30] PonRumbo 90 BL

PonTipo "SANSB PonCL 10 PonTamañoTipo [0.6 0.6]

Rotula "Info

PonTamañoTipo [0.4 0.4]

PonCL 1

SL PonPos [-200 10] BL Rotula [Color y]

SL PonPos [-200 -10] BL Rotula [posicion]

SL PonPos [-200 -30] BL GI 90 botónn GD 90

SL PonPos [-200 -50] BL Rotula [Solo color]

SL PonPos [-200 -70] BL GI 90 botónb GD 90

PonTamañoTipo [0.5 0.5]

SL PonPos [-120 30] BL Rotula [1 _]

SL PonPos [-90 30] BL GI 90 botón 1 GD 90

SL PonPos [-120 10] BL Rotula [2 _]

SL PonPos [-90 10] BL GI 90 botón 2 GD 90

SL PonPos [-120 -10] BL Rotula [3 _]

SL PonPos [-90 -10] BL GI 90 botón 3 GD 90

SL PonPos [-120 -30] BL Rotula [4 _]

SL PonPos [-90 -30] BL GI 90 botón 4 GD 90

SL PonPos [-120 -50] BL Rotula [5 _]

SL PonPos [-90 -50] BL GI 90 botón 5 GD 90

SL PonPos [-120 -70] BL Rotula [6 _]

SL PonPos [-90 -70] BL GI 90 botón 6 GD 90

Fin

;COMENTARIO: Los siguientes procedimientos dibujan los cuadraditos de las jugadas, de las respuestas y de la solución en sus posiciones adecuadas. Las variables CF y CS son contadores para dibujar los cuadrados en las columnas adecuadas.

Para dibujarespuesta

Haz "n :contador1

Haz "b :contador2 - :contador1

Haz "CF 1

Si No :n=0 [dibujanegros]

Si No :b=0 [dibujablancos]

Fin

Para dibujanegros

Repite :n [SL PonPos Lista 55+20*:CF 70-15*:contadordejugadas BL PonRumbo 0 botónn Haz "CF :CF+1]

Fin

Para dibujablancos

Repite :b [SL PonPos Lista 55+20*:CF 70-15*:contadordejugadas BL PonRumbo 0 botónb Haz "CF :CF+1]

Fin

Para dibujasolución

Haz "CS 1

Repite 4 [SL PonPos Lista 145+20*:CS 50 BL PonRumbo 0 botón Elemento :CS :solución Haz "CS :CS+1]

Fin

;COMENTARIO: El siguiente procedimiento permite la posibilidad de volver a iniciar la partida una vez terminada. En caso de no querer volver a empezar el programa se autoelimina de la memoria y ofrece la posibilidad de abandonar LOGO.

Para otravez

PonCT 38

Escribe [¿Otra partida? (S\N)]

PonCT 1

Haz "R LC

Si O :R="S :R="s [MasterMind Alto]

Si No O :R="N :R="n [fallo otravez Alto]

salida

BP BT

RestauraEntorno

BoTodo

Adiós

Fin

;COMENTARIO: El siguiente comando hace que el programa sea autoarrancable una vez interpretado o si el programa es llamado desde fuera del entorno LOGO.

MasterMind

